

# ATK Tähtitieteessä – Harjoitus 2.

## esimerkkiratkaisut

---

Tehtävät 0.-3. ohitetaan tässä. Jos näissä esiintyi ongelmia tai ne herättivät kysymyksiä ota yhteys luennoitsijaan.

4. Plottaa heittoliikkeen ratoja eri lähtönopeuksilla samaan ikkunaan käyttäen eri värejä ja viivatyylejä. Kokeile sekä normaalia *plot,x,y*-komentoa, ja interaktiivista *jokunimi=plot(x,y)*-komentoa. Lisää kuvaajiin akselien nimet ja myös missä yksiköissä akselien arvot ovat. Tallenna interaktiivisesta ikkunasta valmis kuvaaja kotihakemistoosi.

Heittoliikkeen yhtälöt ovat:

$$x = v_x t$$
$$y = v_y t - \frac{1}{2} g t^2.$$

Jossa painovoiman kiihtyvyys  $g = 9.81 \text{ m/s}^2$ , ja alkunopeudet ovat  $v_x$  ja  $v_y$ . Käytä ajan juoksevana numerona sisältävää taulukkoa  $t$ .

### Ratkaisu:

Tehtävässä on syytä lähteä lopusta ja luoda ensin ajan juoksevana numerona sisältävä taulukko  $t$ . Tämä onnistuu komennolla **findgen**. Luodaan lukuja heti tiheävälisesti, eli lisätään jakolasku komennon yhteyteen.

```
IDL> t=findgen(500)/10.  
IDL> help,t  
T FLOAT = Array[500]
```

Eli luotiin liukulukutaulukko muuttujaan  $t$ , joka sisältää arvot  $[0, 0.1, 0.2, \dots, 49.9]$ . Ilman tässä tehtyä kymmenellä jakamista, eli käyttämällä pelkkää komentoa **findgen(500)**, luotaisiin luvut väliltä  $[0., 1., 2., \dots, 499.]$ .

Seuraavaksi voidaan antaa mielivaltaiset x- ja y-suuntaiset alkunopeudet muuttujille  $v_x$  ja  $v_y$ , ja lisäksi määritellään painovoiman kiihtyvyys muuttujaan  $g$ :

```
IDL> vx=6.  
IDL> vy=8.  
IDL> g=9.81
```

Muista että piste lukuarvon lopussa, tai että luvussa on desimaaleja, takaa että muuttuja on liukuluku.


Tämän jälkeen voidaan laskea koordinaatit  $x$  ja  $y$  annetuilla kaavoilla muuttujiin  $x$  ja  $y$ :

```
IDL> x= vx * t
IDL> y= vy * t - 0.5 * g * t^2
```

Voidaan piirtää kuvaaja ensin komennolla `plot` ja aukaistaan sille ensin ikkuna komennolla `window`. Avainsana `/free` avaa ikkunan vapaana olevaan indeksiin.

```
IDL> window,/free
IDL> plot,x,y
```


Huomataan kuitenkin allaolevasta kuvaajasta että arvot menevät nopeasti negatiivisiksi.



Rajataan parametreilla `xrange` ja `yrange` kuvaajaan vain mielenkiintoinen alue. Lisätään myös akseleille selitykset merkkijonoina:

```
IDL> plot,x,y,xrange=[0,10],yrange=[0,10],xtitle='Pituus [m]',ytitle='Korkeus [m]'
```

Jolloin kuvaaja on huomattavasti paremman näköinen:




Muutetaan nyt alkunopeuksia ja piirretään ne aiemmin kuvaajan päälle (`oplot`) eri viivatyyliillä (`linestyle=x`, katso manuaalista x:n arvoja vastaavat tyyliä) ja värillä (`color=x`). Alkunopeuksien muuttamisen jälkeen on aina muistettava päivittää myös taulukot `x` ja `y`. Merkillä `&` voidaan erottaa komentoja samalla rivillä.

```
IDL> vx=5. & vy=7.
IDL> x=vx*t & y=vy*t-0.5*g*t^2
IDL> oplot,x,y,linestyle=2,color=2

IDL> vx=8. & vy=9.
IDL> x=vx*t & y=vy*t-0.5*g*t^2
IDL> oplot,x,y,linestyle=4,color=4
```

Jolloin saadaan alla oleva kuvaaja:



Vastaava voidaan tehdä myös `plot()`-funktioilla. Esimerkiksi:

```
IDL> a=plot(x,y,xrange=[0,10],yrange=[0,10],xtitle='Pituus [m]',ytitle='Korkeus [m]')
```

Jonka päälle voidaan piirtää muilla alkuarvoilla käyttäen `/overplot`-avainsanaa. Huomaa nyt, että parametri `color` ei toimi vastaavasti kuin aiemmin `plot`-komennon kanssa, vaan sille voi syöttää `plot()`-funktiossa suoraan haluamansa värin merkkijonona.

```
IDL> vx=5. & vy=7.
IDL> x=vx*t & y=vy*t-0.5*g*t^2
IDL> b=plot(x,y,linestyle=2,color='red',/overplot)

IDL> vx=8. & vy=9.
IDL> x=vx*t & y=vy*t-0.5*g*t^2
IDL> c=plot(x,y,linestyle=4,color='blue',/overplot)
```

Jolloin on luotu samanlainen kuvaaja kuten aiemmin, ja kuvaaja on helppo tallentaa tässä tapauksessa valikoita käyttäen.